

## S7-226 串型接口采集 DUT-4000 系列模块程序

### 1.主程序

```
//  
//PROGRAM COMMENTS  
//Press F1 for help and example program  
//8 个通道的数据以整型数的形式存于 130 开始的 16 个单元中  
  
NETWORK 1          //Aquisition Module Program  
//  
//NETWORK COMMENTS  
//  
LD      SM0.1  
MOVB    16#09, SMB30  
MOVD    &VB106, VD102  
MOVW    +0, VW100  
ATCH    INT_0, 8  
ENI  
MOVB    5, VB200          //发送命令串  
MOVB    16#23, VB201  
MOVB    16#34, VB202  
MOVB    16#33, VB203  
MOVB    16#30, VB204  
MOVB    16#0D, VB205  
MOVB    0, VB206          //通道计数器  
  
NETWORK 2  
LDN     V100.1  
TON     T33, +30          //300mS 采集一个通道  
  
NETWORK 3  
LD      T33  
LPS  
EU  
LPS  
XMT     VB200, 0          //发送命令  
AB=     VB206, 0  
MOVD    &VB130, VD126    //采集数据结果初值  
LRD
```

```
INCB    VB206
AENO
INCB    VB204                      //ASCII 码的通道号
LPP
AB>=    VB206, 8                  //8 个通道循环
MOVB    0, VB206
AENO
MOVB    16#30, VB204
LPP
=        V100.1

NETWORK 4
LD      V100.0
LPS
AB=      VB101, 9                  //命令字符个数
AB=      VB106, '>'              //接收数据传起始符
AB=      VB114, 16#0D             //接收串结束符
CALL     SBR_0                    //将 ASCII 数据转换成整型数
MOVW     VW124, *VD126
INCD     VD126
INCD     VD126
LPP
MOVB     0, VB101
AENO
R        V100.0, 1
```

## 2.子程序

```
//
//SUBROUTINE COMMENTS
//Press F1 for help and example program
//将接收缓冲区中的 ASCII 码数据转换为整型数，存于 VW124
```

```
NETWORK 1          //将接收的数据转换为整型数
//
LD      SM0.0
MOVD     &VB109, VD116          //数据串指针
MOVW     +0, VW124
FOR      VW120, +1, +5          //转换 5 位 ASCII
MOVB     *VD116, VB123
*I       +10, VW124
```

```
ANDW    16#000F, VW122
+I       VW122, VW124
INCD     VD116
```

```
NETWORK 2
NEXT
```

```
NETWORK 3
LDB=     VB107, '-'
IN VW    VW124                //转换成负数
INCW     VW124
+I       +0, VW124
```

### 3.中断程序

```
//
//INTERRUPT ROUTINE COMMENTS
//Press F1 for help and example program
//接收串处理程序
```

```
NETWORK 1          //接收串程序
//
LD          SM0.0
LPS
MOVB       SMB2, *VD102
INCB       VB101
INCD       VD102
AB=        SMB2, 16#0D          //判断串结束
MOVD       &VB106, VD102
AENO
S          V100.0, 1
LRD
AB>=       VB101, 10           //接收串字符个数限制
MOVD       &VB106, VD102
AENO
MOVB       0, VB101
LPP
CRETI
```